

## An abc Primer

If you can already read and write music, you will find it very easy to learn abc; the language embodies many of the same concepts used in staff notation, and many of the symbols used in abc are deliberately chosen to resemble the equivalent symbol used in staff notation (e.g. the abc symbol for a repeat is :| - it reminds you of the two dots and a barline used for the same purpose in staff notation). If you can't yet read music, abc is an easy place to start; because the two languages have so much in common, you will find that some things you don't understand in one notation will become clear in the other. You can't avoid learning to read conventional staff notation, it's the language which all musicians understand, and abc is not a substitute for that. Some things in abc are much easier for the beginner. In staff notation, the note B is represented by a dot placed on the middle line of a five line staff; a complete beginner spends a lot of time counting up from the bottom of the staff in order to figure out what pitch the symbol represents; in abc the note is simply represented by the letter B, which makes life a lot simpler. On the other hand, abc is much more difficult to sight-read, because it is essentially linear, rather than two-dimensional.

In this file, text in black is intended for everyone. Text in blue is intended for the complete beginner, and if you can already read music you can skip over it.

Pieces of music written in abc are called abc tunes; the word tune is correct here, even if it's a movement of a symphony that we're talking about.

### STRUCTURE OF AN ABC TUNE

An abc tune is divided into two parts, the header and the tune proper. The header contains various pieces of information which are needed to interpret the tune, while the tune proper contains the actual music.

#### The abc header

The header consists of several lines of text, each of which starts with a capital letter followed by a colon. These lines are called fields, and the letter at the beginning is called the field identifier.

The first field in the header always starts with X:, followed by a number. This is the ID field, and the number is usually the number of the tune in the file.

The last field in the header is the key field; it starts with K: and contains a description of the key that the music is in.

In between those two fields there will be several others, their order is not important, but they should always include an M: field which defines the metre, and a T: field which gives the title of the tune.

Fields should always be entered so that the field identifier is hard up against the left-hand margin. BarFly will tolerate empty lines between fields, but other programs may not do so, so you should make a practice of typing abc headers single-spaced.

#### The tune proper

The tune proper starts on the next line after the Key field, and contains symbols representing notes, rests, chords, barlines etc. It may extend over as many lines as necessary, and it ends on the first blank line.

#### A tiny example

This is the shortest tune I can think of:

```
X:1
T:Shave and a haircut
M:4/4
K:C
c2 GG A2 G2 | z2 .B2 .c2 |]
```

From the header we can see that:  
The ID is 1. (It's the first tune in the file)  
Its title is "Shave and a haircut".  
It's in 4/4 time  
It's in the key of C.

The tune proper contains only one line of music, but it is a complete tune, even though it consists of only four notes. You can play this tune (use the Play menu), and switch to Split-Screen mode (in the View menu) to see it displayed in staff notation. Beginners should do this, then scroll down to read the next section.

Music is conventionally written on a set of five horizontal lines, called a Staff. Some people call it a Stave, but the plural is always Staves. The curly symbol at the left is called a clef; there are several types of clef, this one is called a treble clef or G-clef because it specifies that the second line up from the bottom represents the note of G. The heavy black dots represent notes (you can ignore the tails attached to them for the moment). Notes written higher on the staff are higher in pitch. The names of the notes correspond with their position on the staff like this:

f	_____
e	
d	_____
c	
B	_____
A	
G	_____
F	
E	_____

The usual English mnemonic for remembering the notes which correspond to the staff lines is "Every Good Boy Deserves Favours", for EGBDF. Remembering the notes which correspond to the spaces between the lines is easy, since it spells out FACE.

If you are in Split-Screen mode, you can check the way in which the symbols on the staff correspond to the letters in the abc tune by clicking once on each note head. This will cause the note head to be selected, along with the corresponding letter in the abc tune, so you can see exactly which symbols correspond.

## NOTE PITCH

Note pitch in abc is determined by the case of the letter used (for the two octaves above middle C) and by additional symbols outside that range. C means middle C, and the octave above that is written CDEFGAB. At this point, the text switches to lower-case, so the next octave up is written cdefgab. You can go up to higher pitches by adding one apostrophe for each octave to lower-case letters, or down to lower pitches by adding commas to capital letters in the same way. It is illegal to add apostrophes to lower-case letters or commas to upper-case letters. Most music is written in the two octaves above middle C, so in most cases you won't have to type too many commas or apostrophes. If you are writing music for the double bass or piccolo, BarFly offers a couple of shortcuts to save typing, but these are not yet standard abc.

Here's how you write the middle four octaves of the musical scale:

```
X:2
T:Note Pitches
M:none
L:1/4
K:C
```

C,D,E,F,G,A,B, CDEFGAB cdefgab c'd'e'f'g'a'b' c" |]

Beginners should note that the word "Octave" itself is quite confusing, since an octave only contains seven different notes. The first note of the next octave takes you back to the start. It's rather like a foot-rule having thirteen markings on it, despite the fact that there are only twelve inches in a foot. Notes which fall above or below the staff are written on short individual lines called leger lines. The correspondence between these and the note names is easy to remember, as FACE follows directly on from EGBDF and vice versa; if the top line of the staff is f, the next leger line must be a, followed by c' and e'. If the top space is e, then the next space is g, followed by b, d' and f'. The same principle applies going downward.

In abc, you should remember that the symbols change octave whenever you go from B to c (or b to c', or B,, to C, etc.).

In the middle range, the change of case happens above the centre line of the treble staff. (That is also, fortuitously where the tails of the notes change direction.)

## NOTE TIMING

In abc, the duration of notes is determined by a number or fraction which follows the note symbol, so for example A2 is twice as long as A, while A1/2, A/2 or A/ all mean that the note is half the length of A (the last way of writing it is preferred as it takes up least space, but all three are legal abc).

But what does A mean? A single letter written without any modifying number is assigned the default note length, and this is defined in the header by means of the L: field. In the note pitches example above, you will see that there is an L: field in the header which reads L:1/4. This sets the default note length to 1/4, and the notes which I have written below are therefore all represented in the music as quarter notes (or crotchets, if you are British).

The longest note which BarFly can currently represent is the breve (double note in US terminology), and the shortest is the delightfully-named hemidemisemiquaver (or sixtyfourth).

Here they are, in increasing order of length, along with their matching rests (rests are written as z, with the same length rules).

X:3

T:Note Lengths

M:none

L:1/8

K:C

A/8 z/8 A/4 z/4 A/ z/ A z A2 z2 A4 z4 A8 z8 A16 |]

Note that BarFly has no symbol for a breve rest, or for a longa - the next note above the breve. These are not used in modern music, but I may implement them in future.

In general I shall use the more intuitive US names for note lengths - these are their equivalents:

US	British
sixtyfourth note	hemidemisemiquaver
thirtysecond note	demisemiquaver
sixteenth note	semiquaver
eighth note	quaver
quarter note	crotchet
half note	minim
whole note	semibreve
double note	breve

If there is no L: field in the header, the abc standard specifies a fallback for default note length which depends on the metre of the music. If you evaluate the metre as a fraction, the default note is set to 1/8 if the result is 0.75 or greater, and to 1/16 if it is less than this. It's not a good idea to rely on this however, and you should make a habit of explicitly stating what you want the default note length to be by means of an L: field.

Beginners should note that what I'm talking about as "Note Length" here has little to do with the actual duration of time for which a note sounds. What it really means is the time interval between each note and the succeeding one. The actual duration of sound may be much shorter than this (Staccato), or the note may completely fill the time available so there is no silence before the next note (Legato), or anywhere in between.

Note lengths which fall in between these major divisions are represented in staff notation by adding dots, a single dot meaning that the note has one and a half times the length, a double dot meaning one and three quarter times the length, and a triple dot meaning one and seven eighths times. In abc you simply use the appropriate multiplier after the note symbol to get the length you want, so if the default note length is 1/8, A3/2 or A3/ means a dotted eighth, A3 means a dotted quarter, A7/ means a double-dotted quarter and so on. You can actually write any fraction you want after a note; it's legal abc and BarFly will play exactly what you specify. However, some numbers do not translate to staff notation; there is no way to write A5 in staff notation as a single note symbol, and neither BarFly (nor any other abc program as far as I know) will translate this into the pair of tied notes necessary to represent this duration. (You will get an "error in note length" message displayed in the music pane.)

In addition to timed notes you can also have grace notes. These are short notes whose timing is undefined, and up to the player. Grace notes "steal" their time interval either from the following note or the preceding note (In BarFly's player this is an option which you can set in the Player Preferences dialog, along with the actual duration, expressed as a fraction of the default note length.) Grace notes are written in curly braces {a} and represented in the staff notation by an eighth note drawn at half size. Grace note lengths cannot be modified by writing numbers after them.

Here are some more examples:

```
X:4
T:Dotted notes and Gracenote
M:none
L:1/8
K:C
A3/4 z3/4 A7/ z7/ A15/ z15/ A14 z14 A6 z6 {g} []
```

## BARS AND METRE

Music is normally divided into bars, and if the music has a regular beat, each bar contains the same number of beats. The number of beats in each bar, and the length of each beat in musical notes is specified in staff notation by a time signature or metre, which is normally written as a fraction at the left of the first staff. In abc you use the same fraction written into the M: field in the header.

The numerator (top number) of the fraction represents the number of beats in a bar, while the denominator (bottom number) represents the duration of the beat as a musical note. So, for example 3/4 means three beats per bar, with each beat occupying the same length of time as a quarter note. 6/8 means six beats per bar, each beat occupying an eighth note's worth of time. In properly-written music, the sum of the note times in each bar should add up to the value indicated by the metre, so in 4/4 time, however many notes there are in each bar, their total times should add up to a whole note.

Some metres are so common they get abbreviated; 4/4 is also known as "common time" and abbreviated to C. 2/2 is called "cut time" and abbreviated to C|. You can also use these abbreviations in the M: field of an abc tune, and they will be represented appropriately in the music display.

Ordinary bar lines are represented in abc by the vertical slash |, and there are some other special bar line symbols available:

- [| means a thick-thin bar used to mark the start of a section.
- |] means a thin-thick bar used to mark the end of a section.
- || means a double bar used to mark a junction between sections.
- |: is a forward repeat sign, and marks the point to repeat back to.
- :| is a repeat sign, sending the player back to the forward repeat.
- :: is a both-ways repeat, serving as both a forward and backward repeat marker.

BarFly treats the ordinary bar line and the thick-thin as metric: i.e. it will report a timing error if the notes in the bar before don't add up properly. The remaining bars are treated as non-metric (i.e. they may or may not coincide with a metric bar). The two bars on other side of non-metric barlines must add together either to one or two complete bars to avoid a timing error.

In addition to the normal metre settings you can put in the M: field, BarFly also accepts 'none'. In this case, bar checking is turned off, and you can put the bar lines anywhere. This is useful when transcribing music with very frequent metre changes, or songs and slow airs which are performed in free time. It is not part of the abc standard, but many abc programs support it.

Here is a tune with repeat bars, in 6/8 time. If you turn on Highlight note played in the Play menu before playing this tune, you will see how the program deals with the repeats.

```
X:5
T:Statia Donnelly's
M:6/8
R:Jig
K:Em
D|GED G2A|Bge dBA|GED G2A|BGB AFD|
GED G2A|Bge dBA|GEE BAB|GEE E2::f|
g2e fed|ege dBA|gfe afd|Bee e2f|
gfe fed|e/f/ge eBA|GEE BAB|GEE E2:|
```

## MORE ABOUT REPEATS

Sometimes you want the end of a repeated section to play differently on the second time through. To do this you mark the start of the first alternate part with the symbol [1, and the start of the second alternate part with [2. The first alternate part should end in a repeat sign :, and the end of the second alternate part can be marked with || or |].

```
|: <common part> [1 < first time> :| [2 <second time> ||
```

Frequently the alternate part markers coincide with a bar line, and if this is the case you can omit the square bracket, and simply write the number next to the bar line |1. If you do this, there must be no spaces between the bar line and the number. The repeat-from marker at the start is optional; in its absence the repeat will start again from the beginning of the tune, or from the special bar line which marks the end of the previous section, |] or |].

Here's a tune with alternate repeats:

```
X:6
T:Da Scalloway Lasses
L:1/8
M:4/4
Q:1/4=220
```

```

R:reel
O:Shetland
K:Am
eaag egeB | c2 ec gcec | eaag egeB |1 cABG A2 A2 :|2 cABG A2 Ac ||
BE (3EEE c2 ce | d^cde gfed | eaag egeB |1 cABG A2Ac :|2 cABG A4 ||

```

And here's another , where in the second part the two alternate sections start at the beginning of a line. The transcriber has used the [1 format here to avoid putting a redundant bar line at the start of a line:

```

X:7
T:Smash the Windows
C:Trad
R:jig
Q:350
M:6/8
K:D
|:DED F2A | d2f ecA | G2B F2A | E2FGFE |
DED F2A | d2f ecA | Bgf edc |[1 d3 d3 :|[2d3 d2 ]]
f | a2 f d2 e | f2 g agf | g2e cde | e2 f gfe |
[1 a2f d2 e | f2 g agf | g2 e cde | d3 d2 :|
[2 fed gfe | agf bag | fed cde | d3 d3]]

```

There is another way of specifying more complicated playing orders for the parts of a tune, using the P: field, and this is dealt with in the section "More about fields" below.

## ACCIDENTALS AND KEY SIGNATURE

So far we have dealt only with a scale consisting of the seven different notes in an octave (e.g. the white notes on a piano). Such a scale is called a diatonic scale, and its notes are drawn from a set of twelve (if you include the black keys too). If we want to play in different keys we will also have to make use of the black keys, and to represent these we can either indicate that the note below the black key we want is raised in pitch by a half-tone, or that the note above is lowered in pitch by the same amount. We say that the lower note is sharpened by a semitone, or that the upper pitch is flattened. In staff notation, you can mark a note as sharpened by writing a sharp sign (which looks like a hash #) before the note, and as flattened by writing a flat sign (which looks like a stylised letter b). These signs are collectively called accidentals. You can also cancel an existing accidental with a natural sign (for which there is no equivalent in text, but you can see it in the example below).

In abc the following accidentals are available, and should be written immediately before the note symbol (e.g. ^C means C sharp, \_B means B flat):

```

^    Sharp
_    Flat
=    Natural
^^   Double sharp
___  Double flat

```

Accidentals in abc follow the same basic rules as in staff notation:

¥ An accidental applies to the note on which it is placed, and on any other instances of that note (unless cancelled by a natural sign) up to the next bar line.

¥ An accidental applies only to the note on which it is placed, and not to the same note in a different octave. If you

want to place an accidental on a note which occurs twice in a chord in different octaves, you must put it on both notes.

¥ Accidentals are absolute in their effect rather than relative. ^C means that the note is C sharp, even if it was flat, or already sharp before the accidental was applied. This means that you can use "cautionary" accidentals - a mark which is there just to remind the player that the note is still changed, even though it was changed earlier in the same bar, and will have no effect on the pitch.

Here's an example of a Swedish Polska where some of the Fs and Gs are marked as sharp, and where they are not, the transcriber has placed a cautionary natural sign just to show that he really means those notes not to be sharpened.

```
X:8
T:Pennknivsm\ordaren
R:polska J
Z:hn-jp-73
D:V\"asen: Essence
M:3/4
L:1/16
Q:1/4=115
K:Am
abag =fgfd e2e2 | c2ed cBA^G A2E2 | A^GAB c2d2 e2^f2 | d^cd^f aba^g e4 |
abag =fgfd e2e2 | c2ed cBA^G A2E2 | A^GAB c2ce dcBA | ^GABG A8 :|
|: E2c2 E2c2 E2c2 | c2B2 =G2BB B2B2 | cBA^G ABcd e=fed | cdcB ABA^G E4 |
E2c2 E2c2 E2c2 | c2B2 =G2BB B2B2 | cBA^G ABcd e=fed | cBA^G A8 :|
```

If we want to write music in a key other than C or Am (and a few others) we are going to have to flatten or sharpen some notes every time they occur. If we want to use the keys of C# or Abm we will have to place accidentals on every single note! This is clearly a waste of effort, and to avoid doing this we put all the required accidentals together at the start of the staff. This pattern of sharps or flats is called the key signature, and unlike the accidentals, each applies right to the end of the staff (unless changed by an accidental, or by another key signature in the music).

The key signature is one place where abc differs greatly from staff notation. In staff notation the key signature is a collection of sharps or flats which tell you how to pitch the notes, and despite its name it does not tell you what key the tune is in. In abc you state the key and mode explicitly. Here are some examples:

```
K:C          C major - no sharps or flats
K:Am        A minor - no sharps or flats
K:Ab        A flat - four flats
K:E Dor     E Dorian - two sharps
K:D         D Major - two sharps
K:D Maj     D Major - two sharps
K:D Ion     D Major - two sharps
```

The contents of the key field should start with a single capital letter, which may be followed by # or b for sharp or flat. That may be followed by a description of the mode. These modes are recognised by BarFly:

```
Ionian or Major
Aeolian or Minor
Dorian
Mixolydian
Phrygian
Lydian
Locrian
```

It doesn't matter whether the mode is separated from the key by a space or not, modes can be upper- or lower-case,

and can be abbreviated to the first three letters. "Minor" can be abbreviated to the single letter 'm'.

If you play a scale on the piano on the white keys only, and starting on C, you will get the familiar C major scale. If you start on A you will get an A minor scale, and if you start on any other note you will also get a scale, although some of them sound quite strange. Each of these seven scales uses the same key signature (no sharps or flats) but each is on a different keynote. The pattern of tones and semitones in each is distinctive, and these patterns are called modes, and referred to by Greek names.

You will notice that K:E Dor gives exactly the same key signature in the music as K:D. If you enter a tune which is in E Dorian with D in the K: field it will look and sound exactly the same as if you had used the correct K: entry. So why worry about the mode? It's quite clear from the huge number of wrong entries in K: fields in tunes on the net that many people don't care. Where it makes a difference is when you come to harmonise the tune. If you are playing in a session with a guitar player and spring a new tune on him, telling him it's in D will be very misleading - he'll fumble with the chords for a bit until he's figured out that it's not in D at all, but in E. He'd probably be quite happy if you said E Minor, since the Dorian mode uses a minor third, and he'll want to base his accompaniment on the Em chord.

For more information on modes, example tunes and how to figure out what mode tunes are in see this file:

[file:///Modes and Scales](#)

If the conventional key signatures do not fit the music you are working with, you can modify the key signature further by adding global accidentals. The Greek modes are not by any means the only scales possible, and cultures other than Western Europe have invented many variant scales which cannot be represented in this way. To get round this problem, the abc standard allows global accidentals to be placed in the K: field, following on from the key and mode. Global accidentals are written in exactly the same way as normal accidentals, e.g. ^C for C sharp. The abc standard is a little hazy on how global accidentals are to be handled. BarFly treats them by adding the appropriate accidental to the note wherever it occurs in the tune, but they could also be handled by adding them to the key signature.

Here's an example from the Klezmer tradition, using the scale called "Misheberach". This is basically similar to the Dorian mode, but with the fourth sharpened:

```
X:9
T:Der Badchen Freylach
M: 2/4
L:1/16
M:4/4
K: Edor^A
|: eBeg abag | {a}b2{a}b2 gfe2 | eBeg abag |
{a}b4 b4- | bBeg abge | {a}b2{a}b2 gfe2 |fgag fefg | e8 :|
bc'd'c' e'd'c'b | abc'd' b4 | e'd'c'b abc'd' | b4 b4 |
bc'd'c' e'd'c'b | abc'd' b4 |1 bc'd'c' bc'((3gfe) | fge2- e4 :|2 abc'b abgf | e8 ]]
```

There are two special key signatures which can be used in abc to represent bagpipe music. The traditional music of the highland bagpipe is normally written without any key signature. The instrument is diatonic and can only play in one key signature, so it's redundant. In practice the key signature is two sharps (A Mix), but it is not drawn in the music. In addition, beams of timed notes in pipe music are always drawn downward, while gracenote beams are always upward. To force this type of display you use K:HP.

A more rarely-used key signature is two sharps and a natural, and you can get this using K:Hp.

Here's an example:

```
X:10
```



T:The Highlandman kissed his mother

C:MacNeil MS

M:C

%%midi program 1 109

K:HP

B\  
|:{gef}e2 A>e {g}c>e {g}A2 |{g}(3(ce{A}e) {g}A>e {gef}d>B{e}G>B |  
{gef}e2 A>e {g}c/{d}A3/{gf}g>e | {ag}a>f{gf}g>e {gde}d>BG3/{e}B/:|  
{gcd}c2 {gf}g>e a>f{gf}g>e | {gcd}c2 {gf}g>e {gde}d>B{e}G>e |  
{gcd}c2 {gf}g>e a>f{gf}g>e | {gf}g>ag>e {gde}d>B{e}GB :|

BarFly allows several other instructions to be added to the K: field, including clefs, transposition, and the middle directive which changes the pitch of the music as displayed. None of these are yet part of the abc standard, and although all of them are supported by some other programs they are by no means yet universal. You can read about these in this file:

file:///Extensions

## BEAMS

You will have noticed in the last few tunes that some of the note tails are joined together by lines. These are called beams, and they have the visual effect of grouping together notes which fall on the same strong beat, and indicating by their slope which general direction the tune is heading. This makes the music easier to read, but has no effect on the way that it is played. Only notes shorter than a quarter note are joined in this way. The number of 'strong' beats in a bar is not necessarily the same as the number of beats given in the metre. In 4/4 time there are usually only two strong beats, so the bar is divided in two, and if it contains eight eighth notes for example, these will be beamed in two groups of four. In 6/8 time there are normally only two strong beats, so again the bar is divided in two. In 9/8 and 3/4 time the bar will normally be divided into three. Beams can contain rests and gracenotes, and gracenotes themselves can be beamed, but beams of timed notes can never start or end with a rest or a gracenote. The length of beamed notes is indicated by the number of lines joining them to the next note; a single line for an eighth, two for a sixteenth and so on. If the music is a song, you should only beam together notes which fall on the same syllable.

In abc, grouping notes into beams is accomplished by writing them without intervening spaces. Writing notes of a quarter or longer like this has no effect, but the abc is more readable if un-beamed notes are always separated by spaces, regardless of their lengths.

## TIES AND SLURS

Ties and slurs are represented in staff notation by the same symbol, a curved line connecting two note heads together. However, they are really two different things, and abc has separate ways of representing them.

A tie is used between consecutive notes of the same pitch, and indicates that they are to be played as a single note whose length is the sum of the two individual notes. In abc this is written as a hyphen between the two note symbols. There are several reasons for doing this. The first is to create a note of a length which cannot be represented as a single symbol. You can write A5 in abc, and it will play OK, but there is no corresponding symbol in staff notation to draw it. If you write A4-A it will be displayed in staff notation as two separate notes joined by a tie. Another reason for using ties is when you need a note to continue through a bar line, or across a line-break in the music. The hyphen should always be written hard up against the first symbol, so to tie notes across a bar line the hyphen should come before the bar line: a- | a. Yet another reason for tying notes is to preserve the rhythmic beaming. You can tie notes within beams, without beaming, into the first note of a beam or out from the last.

There are examples of all these in the Klezmer tune (#9) above.

In bar 4, there is a b4 tied to a b in the next bar, both to make a note of length 5 and to split it across a bar line.

On the bottom line there is an example of the last note in a beam tied to a longer note, like this:

```
| fge2- e4 :|
```

This could have been written as

```
| fg e6 :|
```

since e6 is a legal dotted note value, but that would have shortened the beam and given the impression of a rhythm change.

A slur is used between consecutive or non-consecutive notes, which may be of the same or different pitches. The symbol used in staff notation is the same curved line (some printers use a heavier line to distinguish ties, but it's not common). In abc slurs are represented using parentheses - (ABc). Slurs are used to indicate that the notes enclosed are to be played legato, or may simply represent phrasing. You can place slurs almost anywhere, across bar lines or line breaks, within or between beams or chords. You can slur grace notes to the next note or the preceding one. Some abc programs automatically join grace notes to the next note by means of a slur. This is quite wrong, as in pipe music grace notes are never written with a slur, and in vocal music and slow airs grace notes are frequently slurred to the preceding note. If you want a slur you must write it like this (`{g}A`) or `{(g)A}` (it doesn't matter whether the opening slur is placed inside or outside the grace note braces).

## TUPLETS

Sometimes in music we want to play a group of notes faster or slower than their nominal time. By far the commonest example of this is the triplet. In this case we want to play three notes in the normal time allotted to two notes, so each of the three notes occupies two thirds of its normal time. Triplets are the twiddly bits of music. In that last sentence, the word "twiddly" is the exact verbal equivalent of a triplet in music. In my accent, at least, the word is pronounced in three syllables: twi-de-ly, but spoken rapidly, so the whole word takes up the same amount of time as the word "music". Triplets are shown in staff notation by writing the numeral '3' above the group of notes, usually with a horizontal line to bracket the three notes together. You can also have groups of numbers other than three, and the general term for this is a tuplet.

Triplets are written in abc by placing an opening parenthesis, along with the number 3, before the three notes - (3ABC, or (3 A B C if the notes are not to be beamed. Note that there is no closing parenthesis. You can write tuplets of other lengths in the same way. Be aware, however, that the rules for determining what is meant by other numbers are quite complicated, and even musical authorities differ in their interpretation. Musicians playing from the score will usually figure out what is meant from the context, but computers need strict rules. These are the rules specified by the abc 1.6 definition:

- (2 - play 2 notes in the time of 3
- (3 - play 3 notes in the time of 2
- (4 - play 4 notes in the time of 3
- (5 - play 5 notes in the time of n
- (6 - play 6 notes in the time of 2
- (7 - play 7 notes in the time of n
- (8 - play 8 notes in the time of 3
- (9 - play 9 notes in the time of n

If the time signature is compound (3/8, 6/8, 9/8, 3/4, etc.) then n is three, otherwise n is two.

These rules are not only very complicated and hard to remember, in practice they always seem to give the wrong answer (for me at least). I always seem to want five notes to play in the time of four, or some other combination which is not available.

Fortunately, abc provides another, and more flexible solution. Instead of placing a single number after the opening

parenthesis, you can write three single digits separated by colons, of the form (p:q:r meaning play p notes in the time of q for the next r notes. So the number of notes in the tuplet is r, and each is played for the fraction q/p of its normal length. The number which appears above the tuplet in the staff display is p, and there are r notes bracketed together by the horizontal line. If r = q then you can omit r (but you still need both the colons). If q follows the standard set of rules as above, it may be omitted, again retaining both colons.

So, for example you could write the standard triplet (3 as (3:2:3, (3:2:, (3::3 or even (3::. If I want five notes to play in the time of four as I usually do, I can write (5:4:5, or just (5:4:.

The notes which follow the tuplet symbol do not have to be all of the same length, and they can be either beamed or un-beamed. You can have gracenotes in tuplets (3A{B}A{G}A (not uncommon in pipe music, where gracenotes appear everywhere). BarFly deals with this situation by modifying the playing length of the gracenotes appropriately, but not counting them in the tuplet.

Here's an example with lots of regular triplets in the second part:

```
X:11
T:Pat Smokes
M:C
L:1/8
C:Martin Hanley
O:Australia
R:Reel
K:G
|:EAAG A2 AG| BAGB AGED| EDEF GFGA| BAGB AGED|
EAAG A2 AG| BAGB AGED| EDEF GFGA| BAGB A4:|
|: eA (3AAA eA (3AAA| GAEA DAEA| eA (3AAA eA (3AAA| (3efg fa gede|
eA (3AAA eA (3AAA| GAEA DAEA| (3efg fa gefd| cABG A4:|
|:ea ~a ba ~a| c'aba aged| eg ~g ag ~g| agbg ageg|
(3efg fd (3efe ed| cABG AGED| EDEF GFGB| AGAB A4:|
```

Here's a case of four notes in the time of two:

```
X:12
T:Donegal Reel
N:as played in Quebec
M:4/4
L:1/8
C:Trad.
R:Reel
K:D
A,C|:D2DE FEFA|dAFA BFAF|E2EF GFGA|BGAF EGFE|D2DE FEFA|
dAFA dfag|fgec dcdB|1AFGE D2 (3A,B,C|2(3ABc ec d2 (3ABc|
|:defg fdAd|fdAd f2ed|cBce gfed|cdef g2 (4:2:fgfe|
defg afd|afdf abag|fgef dcdB|AFGE D2 Ac:|
```

Here's a pentuplet with five notes in the time of two:

```
X:13
T:Mo Chreach A's Mo Dh\`ichairt
H:Equivalent to "alack and well-a-day"
T:My Sorrow and Trouble
B:Joyce, P. W.; "Old Irish Folk Music and Songs"
R:song
```

H:From the singing of Joe Martin of Kilfinane Co. Limerick, 1852.

H:He sang an Irish song to it, of which this is the first verse:

W:Mo chreach a's mo dh'iachairt gan ce'o draoichte air na b'oithribh,

W:A's go siubhalfhainn san o'idhche le'm chroidhe geal na gl'oire.

W:Mo ph'ocaidhe bheith a l'ionadh le geal ph'isidhe cr'oineach,

W:N'a s'asacht s'ud dom'intinn agus l'uidhe s'os le c'obach.

W:

W:Alas and alas, that there is not a fairy-fog on the roads,

W:And that I might walk in the night with thy fair sweetheart of glory.

W:If my pockets were to be filled with white crown pieces--

W:That would not content my mind, and to be married to a clown.

L:1/8

M:3/4

K:G Dor

(3DEF | G2 B2 A3/G/ | G2 F2 (3DEF | G2 B2 A3/G/ | G2 F2 (5:2:5DEFGA |

B2 B2 d3/B/ | c2 c2 d3/c/ | B2 AGF3/A/ | {G}G4 :|

de |f2 edcd | g2 d2 f3/d/ | d2 cA{G}A3/G/ | G2 F2 (5:2:5DEFGA |

G2 F2 d3/B/ | c2 c2 dc | B2 AGF3/A/ | {G}G4 :|

## BROKEN RHYTHMS

It's quite common in traditional music to find long sequences of paired notes where the first note of each pair is dotted, and the second note is halved, or vice-versa. (e.g. A3/B/) This can be very tedious to type, and the resulting abc is quite hard to read. The abc language therefore provides a shortcut. Instead of typing A3/B/ you can place an angle bracket between the two notes: A>B, and instead of A/B3/ you can write A<B. You can think of the angle bracket as transferring half of the time of one note to the other note. An extension of this is that you can use double (A<<B) or triple (A<<<B) angle brackets, which transfer three-quarters and seven-eighths of the time respectively, resulting in the B becoming double- or triple- dotted. In all cases, the total time occupied by the two notes remains the same.

The Strathspey is typical of this kind of dotted rhythm:

X:14

T:The Miller's Rant

C:J Scott Skinner

M:4/4

L:1/8

R:Strathspey

K:D

A,<DF>D A>DF>D|B,<EG>EB>EG>E|A,<DF>D A>DF>D|1 G<BC<A D2 D2:|2 G<BC<A D2 D:|

A|A<df>d a>df>d|B>^de>f g>ec>B|A<df>d a>df>d|B<dc<a d2 d>A|


A<df>d a>df>d|B>^de>f g>ec>B|(3agf (3edc (3BAG (3FED|(3CDE (3A,B,C D2 D|]

There are some situations where you should avoid this construct. Don't use it between unequal notes. If you write A2>B, BarFly will dot the A2 and halve the B, giving A3B/, but the total time occupied is now different from the original. Another program might choose to transfer a fixed amount of time between the two notes giving A5/B/ to keep the total time the same. The abc standard is unclear on this point, so you should avoid the ambiguity by writing the time values out explicitly. A similar situation arises if you use broken rhythm markers in consecutive chains: A>B>C, or where one of the notes is tied A-A>B.

You can have gracenotes attached to broken rhythms A>{g}B, and slurs can start or end on either of the notes without problem.

## CHORDS

You write chords in abc by placing the notes of the chord between square brackets - [CGe]. BarFly permits a maximum of nine notes in a single chord. The abc standard does not say that the notes should be in any particular order, so in general BarFly doesn't care either, except in the following exceptional circumstances. BarFly only plays chords if the Polyphonic item in the Play menu is checked; if that is not the case then it plays only the first-listed note in each chord. There are also some logical limitations to the note order in chords when you use slurs between chords. BarFly will let you slur any combination of notes in a chord to any notes in the following chord, but to do this the notes in the two chords must be in the same order (low to high or high to low), and the slur parentheses will be interpreted in left-to-right order, rather than as mathematical brackets, so the slurred chords [(cA)[dG)] will be joined as:

  
[cA][dG] and not as [cA][dG].

You may come across old tunes which use plus signs '+' instead of square brackets for chords. BarFly will display these tunes correctly, but this comes from an older version of the abc standard which is now deprecated, so you should not use it for new tunes.

Here's a simple example of chords in a concertina arrangement of a fiddle tune:

```
X: 15
T: The Bank of Ireland
R: Reel
M: C
K: G
AB|:[E2c2] AB cBAG|EGDG EGAB|[E2c2]AB cBAG|Add^c [A2d2]AB|
[Ec]AAB cABG|EGDG EGAB|[Ec]AAB cABG|1 Add^c [A2d2]AB:|2 Add^c [A3d3]e||
|:~f3d e2fe|dfed ^cA [E2A2]|~f3d e2fg|afge d2^cd|
ea [e2a2] efge|dfed cAGE| Add^c defg|1 afge [A3d3]e:|2 afge [A2d2]||
```

Chords can be incorporated into beams, and you can use the broken rhythm marker between them [CGe]>[GDB]. You can tie notes into and out of chords, but remember that the tie symbol must go immediately after the note to be tied, not outside the chord, so [CG-e][GDB] is correct to tie the common G between the two chords, while [CGe]-[GDB] is wrong.

Here is a complex example using broken rhythms and beamed chords with ties:

```
X: 16
T: Polska
C: after Adolf Lindman
R: polska J
Z: Transcribed by henrik.norbeck@mailbox.swipnet.se
M: 3/4
L: 1/16
K: D
[DA-]>[FA][FA].[FA] [DA-]>[FA].[FA].[FA] [DA-]>[FA].[FA].[FA]
[D3A3-][FA] .A3.d f4|(5:4:5fgfef .g2.f2 .e2.d2|.c2.B2 (AB).A.G (FG)FE|
[DA-]>[FA].[FA].[FA] [DA-]>[FA].[FA].[FA] [DA-]>[FA].[FA].[FA]
[D3A3-][FA] .A3.d f4|(5:4:5fgfef .g2.f2 .e2.d2|(cd).e.c d4 z4:|
|:(Ad).f.d (Ad).f.d (Ad).f.d|(Ac).e.c (Ac).e.c (Ac).e.c|
(DG).B.G (DG).B.G (DG).B.G|(DF).A.F (DF).A.F (DF).A.F|
```

(D3F) (A2d2) (f2a2)|(5:4:5aba^ga (b2a2) (g2e2)|[e3c3]d [d4D4] z4:|

and here's an example which uses tied notes between chords to represent multiple voices playing simultaneously, and multiple slurs. This would be much easier to do in multi-voice abc, but this is an old transcription I made before multi-voice abc was available. Note that the use of the transpose specification in the K: field is not part of the abc standard but is supported by some other programs.

X:17

T:St Louis Tickle

C:Theron Bennett

R:Ragtime

H:Piano rag arranged for guitar

H:I believe it was written for the St Louis World's Fair of 1904.

Guitar arrangement by Happy Traum, but this is transcribed from my own playing, and since I have played it for many years it may have been a little 'folk processed'.

M:12/8

Q:300

K:C Transpose -12 %Guitar music is played an octave lower than written

P:A

```
[B3G,3] [c3A,3] [d3B,3] |:\n[c3C3] [(g3(e3(G,3) [(^f3) (^d3)(A,3) [=f3]=d3)B,3)] | [(e2C2)^d[e3]E3][c3-G,3][c3E3] | [(d2F,2)c[A3]F3] [(d2^F,2)c[A3]F3] | [(e2C2)^d [e3]E3][c3-G,3][c3E3] | [(d2F,2)c[A3]F3] [(d2^F,2)c[A3]F3] | [(e2C2)^d[e3]E3] [g3-^c3-A3-A,3][g3c3A3E3] [1 [(d2^F,2)^d[e2]D2](=d[^d2F,2]e)(=d2D2)^d | [(e2E2)c[A3]D3] [g3-B3-G,3][g3B3B,3] :| [2 [(e2^F,2)c[d3]A3] [(e2G,2)^d[=d3)A,3] | [c4G4E4C4] z4-z ]
```

P:B

```
E,3 | : F,3[(f3F3)][(e3)F,3][d3)F3] | [(c2F,2)B[c3)F3][(d2F,2)c[A3)F3] | F,3[(f3F3)][(e3)F,3][d3)F3] | [(c2F,2)B[c3)F3][(d2F,2)c[A3)F3] | [1 G,3[(g3c3^A3E3)[(^f3)c3A3G,3][=f3)c3A3E3] | [(e2C2)^d[e3]E3][g3-^c3-A,3][g3c3E3] | [(d2^F,2)^d[e2]D2](=d[^d2F,2]e)(=d2E2)^d | [(=e2^F,2)c[A3)D3] [g3-G,3] [g3E,3] :| [2 [(d2^G,2)B[d3)F3][(f2^G,2)e[d3)F3] | [(c2F,2)B[c3)F3] [(d2^F,2)c[A3)F3] | [(a2G,2)f[g3)G3] [(a2G,2)^g[=g3)G3] | [f6c6A6F,6] F,3 ^F,3 ]
```

P:C

```
G,3[(b3f3D3)[(^a3)f3G,3][=a3)f3D3] | [^g2f2G,2][gf]D3 [=g3-f3-G,3] [g3f3B,3] | C3 [c'3g3c3][b3C3]c3 | [(a2C2)ec3] [g3-C3][g3c3] |[1 [(g2G,2)d[e3)D3] [(f2G,2)e[d3)D3] | [(g2A,2)c[d3)E3] [(e2E,2)d[c3)E3] | [(d2^F,2)^c[d3)D3] [(e2^F,2)f[^f3)D3] | [g3-B3-G,3] [g3B3D3] [f3-B3-G,3] [f3B3D3] :|[2 [(c2^F,2)A[c3)^D3][(^d2F,2)=d[c3)D3] | [(e2C2)^d[e3)E3] [(g2A,2)e[^c3)E3] | [(e2^F,2)c[d3)D3] [(e2G,2)^d[=d3)D3] | [c6G6C6] z3 ]
```

P:Coda

```
E,3 | [(A3F,3)[c3)F3] [d3-^F,3][d3)F3] | [(e2C2)^d[e3)E3] [g3-A,3][g3A3] | (g2ag2^d e2cA2G) | (e2^d=d3) [c6G6C6] ]
```

## GUITAR CHORDS AND ANNOTATIONS

Guitar chord symbols can be added to abc tunes by writing them in double quotes immediately before the note to which they apply, e.g. "Am7"G. BarFly can draw the chord symbols either above or below the staff (it's an option in the Viewer Preferences dialog). BarFly does not play guitar chords (although some other programs do); it will, however, transpose them correctly.

Here's an example:

X:18

T: The Machine without Horses  
R: Jig  
Z: John Chambers <jc@trillian.mit.edu>  
M: 6/8  
L: 1/8  
S: Adaptations and guitar chords by Robin Shaylor 2002  
K: G  
|: D \\  
"G" G3/2A/B/c/ dBG | Gge dBG | "C" c>ec "G" dBd | "Am" ABG "D7" FED |  
"G" G3/2A/B/c/ dBG | "Em" Ggf efg | "D" a>fd "A7" ge^c | "D" d3- d2 :|  
[| d \\  
"G" d>Bd dBd | "D7" cAc cAc | "G" B>dg dBG | "D7" ABG FED |  
"G" d>Bd dBd | "D7" cAc cAc | "G" Bdg dBG | "D7" c>AF "G" G3 |  
"G" d>Bd dBd | "D7" cAc cAc | "G" B>dg "Em" dBG | "Am" ABG "D7" FED |  
"G" G>DG BGB | "Em" dBd g2d | "Am" e>fg "D7" a>gf | "G" g3- g2 ||

Lots of abc users have used the guitar chord mechanism to write text annotations into the music. This is not a good idea for those programs which play or transpose guitar chords, and there is a simple extension to abc which gets around this problem. If you write text between double quotes starting with one of the characters ^, \_, <, > BarFly (and some other programs) will not treat it as a guitar chord, but as pure text. If you use ^, the text will be placed above the note to which it applies, and if you use \_ it will be placed below. The angle bracket markers can be used to display text to the left or right of the note head, useful for adding fingering to chords.

## ACCENTS

Accents are marks applied to a note in the staff display to indicate to the player that it should be played in a particular way. Accents are represented in abc by a single character placed immediately before the note to which they apply. These are the accents listed in the abc 1.6 standard:

**Staccato** Represented by a period, e.g. .A in the abc, and by a dot placed over or under the note head in staff notation. A staccato mark causes the physical length of a note to be shortened while leaving its musical length (i.e. the time until the next note) unchanged. In BarFly you can change the physical length of staccato notes by moving the staccato scrollbar in the player preferences.

Here's an example (staccato triplets are a characteristic of Donegal fiddle music):

X:19  
T:Gan Ainm  
R:Reel  
M:4/4  
K:Gm  
DG(3.G.G.G BdcB | AF~F2 FdcA | DG(3.G.G.G (3Bcdfd | {cd}cAfc AGGF |  
DG(3.G.G.G ~B3c | AF~F2 FdcA | DG(3.G.G.G (3Bcdfd | {cd}cAfc AGG2 ||  
Ggga b2ag | fcac gcfc | ~g3a (ab)ag | fd{cd}cA BGG2 |  
zgga b2ag | fcfg agfa | ~g2fg dgcg | AFcA (FG2)F ||

**Bowing marks** represented by the letters u and v in abc for up-bow and down-bow respectively, and by special symbols written over the note head in staff notation. These have no effect on playing.

Here's an example:

X:20

T:Untitled Polka  
C:from John W. Summers  
Z:Transcribed by Joel Shimberg  
M:C|  
L:1/8  
Q:375  
K:G

```
vduc | vBuG u[G2B2] [G2vB2] GG | uEuC u[C2E2] v[C2E2] (3EFE | DEFG ABcA | BG [G2B2]"a"[G2B2] :|  
Bc |: d2B2 g4 | (3fgf e2 ([e2c2] [e2e2]) | e2 de dAFG | A2G2 G2Bc |  
d2B2 g3 B | c2A2a3g | f2 AB cAFA|1 G6 Bc :|2 [B2g2] [gB] [gB] [g2B2] ||
```

Tilde (~) In abc this symbol represents an unspecified ornament, to be interpreted according to context. It is represented in the staff notation by a turn symbol (which looks much like a tilde), but a classical turn is only one possible interpretation. Depending on context, it may signify a fiddle roll or piper's crann or whatever you wish. BarFly has a system of macros which can be used to determine what gets played here. For more information on macros read this file:

file://:Using BarFly Macros

The abc standard also reserves the capital letters H to Z for applications to use locally to represent various other musical symbols. Some of these (T, M, P, & L) have been adopted by several different programs and are more or less standard. Others are completely unstandardised. This is the list of available symbols in BarFly:

T	Trill
M	Mordent
P	Lower Mordent (Pralltriller)
L	Emphasis mark (or vertical episema in Gregorian notation)
H	Fermata
I	Inverted fermata
S	Short phrase mark
R	Medium phrase mark
Q	Long phrase mark.
K	Horizontal episema (in Gregorian notation only)
W	Segno

X:21  
T:Accents and ornaments  
M:none  
K:C

```
~A TA MA PA LA HA IA SA RA QA WA []  
w: tilde T M P L H I SÊÊÊÊÊ RÊÊÊÊÊ QÊÊÊÊÊ W
```

BarFly also allows you to assign the symbols to different letters, and also has some additional symbols available, Redefinable symbols are an extension to abc which is not yet supported by any other programs, and you should read this file for more information:

file://:Extensions

Some of these symbols are drawn either above or below the note head, depending on which way up it is (~, L), others are always placed above the staff (T, M, P, H, I & W) and the three phrase marks are placed in between the note to which they apply and the previous note.

The two fermata are played (they modify the length of the note to which they are attached) and you can set the amount by which they lengthen or shorten the note in the Player Preferences dialog. The trill and the two mordents can be



played by means of macros, and there are suitable macro definitions under "Mordents and Trills" and "Bach ornaments" in the Global Macro File sub menu, near the bottom of the Edit menu.

## MORE ABOUT FIELDS

You have probably noticed that there are many other fields available for use in the tune header, in addition to the essential ones which I introduced at the beginning. Some of these are not used at all by BarFly, and generally BarFly will simply look in the header for the information it needs and ignore any additional material. Some fields can also be used in the tune proper, and I'll illustrate that later. Meanwhile, these are the fields which BarFly does use:

### ID field (X:)

This is essential, can only occur once in a tune, and must be the first field in the header. The content of the field is an integer number, which for BarFly must be in the range 0 - 32768.

### Title field (T:)

BarFly treats this as optional. It can occur multiple times in the header and in the tune. The first instance of the T: field in the header contains the main title, and is displayed centred at the top of the displayed music. The main title is also displayed in the index. The contents of subsequent instances of the T: field in the header contain alternative titles and are concatenated with commas and displayed in smaller type, centred below the main title.

### Composer field (C:)

An optional text field permitted only in the header, whose contents are displayed in italics at top right of the displayed music.

### Default note length field (L:)

An optional field which may occur once in the header, and multiple times in the tune proper. This field contains a number (or more usually a fraction) which defines the length of an unqualified note in the tune proper. So, if the field contains L:1/8, then a single note letter means an eighth note.

If there is no L: field in the header then a default value is used based on the content of the M: field. If the metre is evaluated as a fraction, a value of 0.75 or greater (e.g. 3/4, 4/4, 6/8) will give a default note length of an eighth, while if less than 0.75 (e.g. 2/4, 3/8) the default is a sixteenth.

### Metre field (M:)

This is an essential field in the header, where it should occur only once. It may also appear multiple times in the tune proper.

### Key field (K:)

An essential field in the header, where it not only provides information about the key signature, but also marks the position of the end of the header. The K: field can also occur multiple times in the tune proper.

### Rhythm field (R:)

An optional field which may occur once in the header, and multiple times in the tune proper. It contains a text description of the rhythm in which the tune is to be played, e.g. jig, reel, march etc. BarFly uses this field in conjunction with the metre field to determine which stress program to apply when playing the tune. To see what rhythms are recognised you should read the Stress Programs file itself, which also contains information on how you can create your own stress programs. If stress programming is disabled in the Player Preferences dialog, or the contents of the R: and M: fields do not match, the field will be ignored.

### Tempo field (Q:)

An optional field which may occur once in the header, and multiple times in the tune proper. The abc 1.6 definition gives three different formats for the contents of this field.

A simple number means that that number of default notes will play per minute. If the default note length is 1/8, then

Q:300 means play 300 eighth notes per minute.

If you want to state the tempo in terms of a multiple of the default note length, you can write e.g. Q:C3 = 40. In this case "C" means the default note length, "C3" means three times that. If the default note length is 1/8, then the whole thing means play forty dotted quarter notes per minute.

Finally you can specify an absolute tempo independent of the default note length, by writing e.g. Q:1/4 = 100, meaning play 100 quarter notes per minute.

Of the three formats, the first is commonly used, because it's the simplest to type. I have never seen an abc tune which used the second format. The third format is the one which is most meaningful to musicians, because you don't have to figure out what the default note length is in order to understand it.

BarFly can take its setting of the tempo from three different sources. If there is a Q: field in the current tune, in any of the three formats, it will use that. If there is no Q: field, but there is a valid R: field it will take its tempo from the indicated stress program, provided that stress programming is enabled. If neither of these apply, it will use the default tempo set in the Player Preferences dialog. You can also override the Q: field in the tune by means of a checkbox in the same dialog, in which case the Q: field will be ignored, and the tempo used will be taken from the R: field or will be the default tempo if that does not apply.

#### Part field (P:)

An optional field which may occur once in the header, and multiple times in the tune proper. This field has two distinct and separate uses.

If there is no part field in the header, then any part fields in the tune proper are interpreted as simple labels marking the start of a section of the music. In this case, the content of the field can be any string of characters, e.g. P:Introduction.

If there is a part field in the header, it is used to determine the order in which the labelled parts are played. In this case, the part fields in the tune proper must contain only single capital letters A B C etc. while the part field in the header contains a string of letters defining the playing order, e.g. P:AABAAC. BarFly (and most other abc programs) will also permit the use of numbers and parentheses to define complex playing orders, e.g. P:A(BCD)3A, and will tolerate the use of spaces here.

#### Words (W:)

According to the abc 1.6 definition, lyrics in the W: field are permitted only in (or following) the tune proper. Some programs also permit W: fields to be used in the header. BarFly does not object to this but will not display the lyrics in this case. The W: field is most commonly used to add multiple verses after the tune proper. You can also place single or multiple lines of words between the lines of music in W: fields, however, the text is reproduced verbatim, and will not be aligned with the notes. BarFly (and some other programs) also support the w: field (in lower case) for words to be aligned with the notes. This is an extension to abc, and is described here:

file://:Extensions

The following fields are described in the abc 1.6 standard, and are either information-only fields or are meaningless within the context of BarFly, and are therefore ignored:

A:	Area	e.g. Shropshire
B:	Book	e.g. A.L.Lloyd, Folksongs of England
D:	Discography	e.g. Malicorne "Almanach"
E:	elemskip	I have no idea - it's only meaningful to abc2mtex, the original abc program.
F:	File name	Used by the index program which comes with abc2mtex.
G:	Group	ditto

H:	History	e.g. This tune was stolen from the fairies by my great great grandfather... The standard specifically says that this field can continue over more than one line without repeating the field identifier on each line.
I:	Information	Used by playabc (archaically) to set tempo, before the Q: field was introduced.
N:	Notes	Absolutely anything you want to write about the tune.
O:	Origin	e.g. Irish
S:	Source	e.g. learned from Jimmy McBeath
Z:	Transcription note	The conventional place to put the transcriber's name, email address or URL.

BarFly also uses two fields which are not part of the abc standard. The lower-case m: field is used for macro definitions and the U: field is used to reassign accent symbols. Read the Extensions file for more information.

## FIELDS IN THE TUNE PROPER

Many of the fields listed above can be used in the tune proper. In the abc 1.6 standard, fields can only be used with the field identifier up against the left margin, so to introduce a key or metre change in the middle of a line of music you must break the line of text. To avoid introducing an unwanted line break in the music you should use the backspace continuation character \ at the end of the previous line to indicate that the line continues below.

Here's an example of a rhythmically-complex tune with lots of metre changes:

```
X:22
T:The Trowie Burn
O:Shetland
M:6/8
C:Fredemann Stickle
K:A Dor
G2A Bed|BAG A2c | (2{Bc}BA (2:3:3AG/A/ |{Bc}B2A (2G<E |
G2A Bed|BAG Bcd | (2eE {Bc}B2A | (2G<E E3 :|
g2a baf \
M:9/8
g2d (2B<G A2c \
M:6/8
(2{Bc}BA (2:3:3AG/A/ |
{Bc}B2A (2G<E | g2a baf \
M:9/8
g2d (2B<G Bcd |
M:6/8
(2eE {Bc}B2A | (2G<E E3 \
g2a baf \
M:9/8
g2d (2B<G A2c \
M:6/8
(2{Bc}BA (2:3:3AG/A/ |
{Bc}B2A (2G<E | G2A Bed|BAG Bcd | (2eE {Bc}B2A | (2G<E E3 ||
```

In a tune with lots of in-tune fields, this breaks up the abc and makes it quite hard to read. Most modern abc programs offer an alternative inline format for fields. Here the field is written in square brackets [M:6/8], and need not be at the left hand margin. The opening bracket, field identifier letter and colon must all be written without intervening spaces.

Here's the same tune again, written using inline fields:

```
X:23
T:The Trowie Burn
```

O:Shetland

M:6/8

C:Fredemann Stickle

K:A Dor

```
G2A Bed|BAG A2c | (2{Bc}BA (2:3:3AG/A/ |{Bc}B2A (2G<E |
G2A Bed|BAG Bcd | (2eE {Bc}B2A | (2G<E E3 :|
g2a baf |[M:9/8] g2d (2B<G A2c |[M:6/8](2{Bc}BA (2:3:3AG/A/ |
{Bc}B2A (2G<E | g2a baf |[M:9/8] g2d (2B<G Bcd |
[M:6/8] (2eE {Bc}B2A | (2G<E E3 |g2a baf |[M:9/8] g2d (2B<G A2c |[M:6/8](2{Bc}BA (2:3:3AG/A/ |
{Bc}B2A (2G<E | G2A Bed|BAG Bcd | (2eE {Bc}B2A | (2G<E E3 ||
```

If you switch back and forth between the two tunes you will see that they produce an identical music display, but the second version is easier to work with as the lines of text now correspond with the staves.

Where you have multiple fields adjacent to one another in the tune proper, you should pay attention to the order in which they are placed, and this is particularly the case where one of the fields is a P: field. It is quite common to find a tune part which starts with a key or metre change, and in this case the P: field must precede the other fields. This is logically necessary, since the new part starts at the P: field, and if you were to place the key or metre fields before this they would be located at the end of the previous part, which is not the same, particularly if there is a P: field in the header which dictates that the parts are not to be played consecutively.

Where there is a simultaneous change of metre and rhythm, the M: field should precede the R: field, so that the new metre is in place before the program checks that the specified rhythm matches it.

Even if the default note length was originally set by the metre field in the header, a change of metre in the tune proper will not change it again. If you need to change the default note length in the tune proper you must insert an L: field.

## FILE AND SECTION HEADERS

The file header is the part of an abc file between the beginning and the first tune. The abc standard permits you to write fields here, outside of any tune, and these fields will be applied to all the tunes in the file. BarFly recognises M:, L: and Q: fields here, as well as macro and symbol definitions (m: and U:). These fields are only parsed when the file is being opened, so if you edit this part of the file you will need to close and re-open it to see any difference. If the fields in question are repeated in a tune header, the definition in the tune header will override that in the file header for that tune only.

In general, it's a bad idea to use file headers to store important information. There is a strong tendency to forget that the definitions are there, and then later you may copy a tune from the file and put it elsewhere without some vital piece of information. It's better to keep your tunes self-contained.

Section headers are the parts of an abc file between tunes. The abc standard also permits you to place fields here, to apply from this point on in the file. This only makes sense to a program like abc2mtex (the original abc program) which reads files through linearly from beginning to end, translating each tune to another format as it does so. In a program like BarFly, which treats the file as random-access, allowing you to jump about between the tunes in any order, the use of section headers would slow things down immensely, as the program would have to search the whole file from the beginning up to the selected tune every time you clicked on the index. Section headers are (I think) the only feature of the abc v1.6 standard which BarFly does not implement.

## MISCELLANEOUS

The percent sign % is used in abc to mean "ignore the rest of this line". BarFly will let you use it almost anywhere to add comments which are only to be read in the abc source.

The backslash \ can be used at the end of a line of text to signify that the line continues on the following line without a line-break in the music. The electronic transmission of text files with long lines is a somewhat insecure process, as many of the programs which handle files on their way around the internet have the habit of inserting line breaks into them. In abc this may cause serious disruption, as many of the combination symbols are composed of characters which must be written consecutively. To avoid this you should keep lines short, and make use of the backslash on line where you do not intend a line break. From this point of view, the first version of The Trowie Burn (X:22 above) is to be preferred to the second (X:23), even though it's less readable. If all of the lines of a tune (or all except the last) are continued in this way, the program will not try to display the tune all on one line, but instead will wrap the lines to get a display with a reasonable note density. However, the line breaks will not necessarily coincide with the ends of parts, and it's probably not a good practice to rely on this.

## NOW WRITE YOUR OWN!

If you haven't already tried it, now's the time to enter some music of your own. Use the New command to open a new window and type in a simple abc header, with the appropriate title, metre and key fields. You should now have something like this:

```
X:24
T:My Tune
M:4/4
K:C
```

Now you can switch the window to Split-Screen mode, and the program will display "No Tune!". Place the insertion point on the last line of the tune, and start typing; the notes will appear as you type. When you reach the end of a line, start the next line with a barline, and move the insertion point to before it. Type a few notes. If you have the "Allow last stave to be un-justified" option on in Viewer Preferences, the staff will grow from the left as you type. Sometimes the program won't display anything until the symbol you are typing is complete; sometimes it will display an error message. Just keep going and all will be well. Once you have typed a few bars play the tune and see if it sounds right. If the error icon changes to a frown there's an error of some kind; click on the error icon to read the error message.

Have fun!

-----  
-----  
Back to the table of contents:

file:///Table of Contents